

SIEMENS



FAQ • 09/2014

Diagnosedaten im Anwenderprogramm auswerten

S7-300 / S7-400 CPU

<http://support.automation.siemens.com/WW/view/de/101906697>

Dieser Beitrag stammt aus dem Siemens Industry Online Support. Es gelten die dort genannten Nutzungsbedingungen (www.siemens.com/nutzungsbedingungen).

Security-hinweise

Siemens bietet Produkte und Lösungen mit Industrial Security-Funktionen an, die den sicheren Betrieb von Anlagen, Lösungen, Maschinen, Geräten und/oder Netzwerken unterstützen. Sie sind wichtige Komponenten in einem ganzheitlichen Industrial Security-Konzept. Die Produkte und Lösungen von Siemens werden unter diesem Gesichtspunkt ständig weiterentwickelt. Siemens empfiehlt, sich unbedingt regelmäßig über Produkt-Updates zu informieren.

Für den sicheren Betrieb von Produkten und Lösungen von Siemens ist es erforderlich, geeignete Schutzmaßnahmen (z. B. Zellenschutzkonzept) zu ergreifen und jede Komponente in ein ganzheitliches Industrial Security-Konzept zu integrieren, das dem aktuellen Stand der Technik entspricht. Dabei sind auch eingesetzte Produkte von anderen Herstellern zu berücksichtigen.

Weitergehende Informationen über Industrial Security finden Sie unter <http://www.siemens.com/industrialsecurity>.

Um stets über Produkt-Updates informiert zu sein, melden Sie sich für unseren produktspezifischen Newsletter an. Weitere Informationen hierzu finden Sie unter <http://support.automation.siemens.com>.

Inhaltsverzeichnis

1	Einleitung	3
2	Diagnosealarm aktivieren und Diagnosealarm-OB (OB82) im Anwenderprogramm der CPU einfügen	4
3	Anwenderprogramm.....	8
3.1	Übersicht	8
3.2	Diagnosealarm-OB (OB82	8
3.3	OB1	10
3.4	FB30 „FB_chaF“	10
3.4.1	Diagnosedaten mit der Systemfunktion SFC51 „RDSYSST“ auslesen.....	10
3.4.2	Aufbau der Datenstruktur „SZL_HEADER“	12
3.4.3	Aufbau der Diagnosedaten	12
3.4.4	Drahtbruch auf Kanal 0 und/oder Kanal 1 einer digitalen Eingabebaugruppe erkennen.....	15
3.4.5	SFC52 „WR_USMSG“	16
3.5	FB20 „FB_extF“	17
3.6	FB10 „FB_intF“	17

1 Einleitung

In diesem Dokument wird beschrieben, wie Sie bei diagnosefähigen Baugruppen eine Änderung des Diagnosezustands im Anwenderprogramm der CPU auswerten und darauf entsprechend reagieren können.

Wenn eine diagnosefähige Baugruppe, bei der Sie den Diagnosealarm freigegeben haben, eine Änderung ihres Diagnosezustands erkennt, stellt sie eine Diagnosealarmanforderung an die CPU:

- Es liegt eine Störung vor oder eine Komponente muss gewartet werden oder beides (kommendes Ereignis).
- Es liegt keine Störung mehr vor und keine Komponente muss mehr gewartet werden (gehendes Ereignis).

Daraufhin ruft das Betriebssystem der CPU den Diagnosealarm-OB (OB82) auf.

Wenn ein Diagnosealarm ausgelöst wird, trägt die gestörte, diagnosefähige Baugruppe automatisch 4 Byte Diagnosedaten und ihre Anfangsadresse in die Startinformation des Diagnosealarm-OB (OB82) und in den Diagnosepuffer ein. Damit erhalten Sie die Information, auf welcher Baugruppe und wann ein Fehler aufgetreten ist.

Weitere Diagnosedaten der gestörten, diagnosefähigen Baugruppe (auf welchem Kanal der Fehler aufgetreten ist, um welchen Fehler es sich handelt) können Sie mit einem entsprechenden Programm im Diagnosealarm-OB (OB 82) auswerten.

Mit der Anweisung „RDSYSST“ können Sie Baugruppendiagnosedaten auslesen und mit der Anweisung „WR_USMSG“ diese Information in den Diagnosepuffer eintragen.

Hinweis

In den technischen Daten der Baugruppe finden Sie die Information, ob die Baugruppe Diagnosefunktionen unterstützt und die Diagnoseinformationen auslesbar sind.

2 Diagnosealarm aktivieren und Diagnosealarm-OB (OB82) im Anwenderprogramm der CPU einfügen

Dieses Beispiel zeigt, wie Sie die Diagnosedaten einer digitalen Eingabebaugruppe, z. B. SM321 (6ES7321-7BH01-0AB0) im Anwenderprogramm der S7-300 CPU auslesen, auswerten und visualisieren können.

Übersicht

Die folgende Abbildung zeigt die Hardware-Konfiguration des Beispielprogramms. Das Beispielprogramm ist mit STEP 7 Professional V13 erstellt.

Abbildung 2-1




Tabelle 2-1

Nr.	Komponente	Artikelnummer	Hinweis
1	CPU315-2 PN/DP	6ES7315-2EH14-0AB0	Alternativ können Sie jede beliebige S7-300 CPU einsetzen.
2	SM321, 16DE, DC24V	6ES7321-7BH01-0AB0	Alternativ können Sie jede beliebige Eingabe- und Ausgabebaugruppe einsetzen, die diagnosefähig ist.

Hinweis


In folgendem Handbuch finden Sie die Information, welche S7-300 Baugruppen Diagnosefunktionen unterstützen.

<http://support.automation.siemens.com/WW/view/de/8859629>

Diagnosealarm in der digitalen Eingabebaugruppe aktivieren


In der Projektnavigation doppelklicken Sie im Geräteordner der S7-300 CPU auf den Eintrag „Gerätekonfiguration“. Im Arbeitsbereich wird der Hardware- und Netzwerkeditor geöffnet.

Abbildung 2-2



Im Hardware- und Netzwerkeditor öffnen Sie in das Register „Gerätesicht“. Die Konfiguration der S7-300 wird angezeigt.

Abbildung 2-3




Markieren Sie digitale Eingabebaugruppe. Im Inspektorfenster werden die Eigenschaften der digitalen Eingabebaugruppe angezeigt.

Im Register „Allgemein“ navigieren zu „Eingänge > Kanal 0 – 7“. Aktivieren Sie für die Kanäle 0 bis 7 den Diagnosealarm für den „Drahtbruch“.

Im Register „Allgemein“ navigieren zu „Eingänge > Kanal 8 – 15“. Aktivieren Sie für die Kanäle 8 bis 15 den Diagnosealarm für den „Drahtbruch“.

2 Diagnosealarm aktivieren und Diagnosealarm-OB (OB82) im Anwenderprogramm der CPU einfügen


Abbildung 2-4



Diagnosealarm-OB (OB82) im Anwenderprogramm der S7-300 CPU hinzufügen

In der Projektnavigation doppelklicken Sie im Programmbaustein-Ordner der S7-300 CPU auf den Eintrag „Neuen Baustein hinzufügen“. Der Dialog „Neuen Baustein hinzufügen“ wird geöffnet.

Abbildung 2-5




Klicken Sie auf die Schaltfläche „Organisationsbaustein“ und wählen Sie unter „Fault interrupts“ den Diagnosealarm-OB (OB82) aus.

2 Diagnosealarm aktivieren und Diagnosealarm-OB (OB82) im Anwenderprogramm der CPU einfügen

Klicken Sie auf die Schaltfläche „OK“. Der Diagnosealarm-OB (OB82) wird im Anwenderprogramm der S7-300 CPU eingefügt.


Abbildung 2-6



3 Anwenderprogramm

3.1 Übersicht

Die folgende Abbildung zeigt eine Übersicht des Anwenderprogramms.
Abbildung 3-1




3.2 Diagnosealarm-OB (OB82)

Für die Auswertung der Diagnosedaten im Diagnosealarm-OB (OB82) wird die Modul-Adresse der digitalen Eingabebaugruppe benötigt.

Öffnen Sie die Eigenschaften der digitalen Eingabebaugruppe, um die Modul-Adresse zu ermitteln. Im Register „Allgemein“ navigieren Sie zu „E/A-Adressen“. Die Anfangsadresse der Eingänge entspricht der Modul-Adresse. In diesem Beispiel ist die Anfangsadresse der Eingänge und somit die Modul-Adresse 4.

Abbildung 3-2



Im Programm des Diagnosealarm-OB (OB82) wird ausgewertet, ob der Diagnosealarm durch einen internen Fehler oder externen Fehler der digitalen Eingabebaugruppe ausgelöst wurde.

Wenn ein interner Fehler vorhanden ist, ist die Variable „OB82_INT_FAULT“ des Diagnosealarm-OB (OB82) auf „1“ gesetzt. Daraufhin wird im OB82 der Merker M20.1 „internal_failure“ auf „1“ gesetzt. Wenn der M20.1 „internal_failure“ auf „1“ gesetzt ist, wird der FB10 „FB_intF“ im OB1 aufgerufen und ausgeführt.

Wenn ein externer Fehler vorhanden ist, ist die Variable „OB82_EXT_FAULT“ des Diagnosealarm-OB (OB82) auf „1“ gesetzt. Daraufhin wird im OB82 der Merker M20.2 „external_failure“ auf „1“ gesetzt. Wenn der M20.2 „external_failure“ auf „1“ gesetzt ist, wird der FB20 „FB_extF“ im OB1 aufgerufen und ausgeführt.

Ein Kanalfehler ist ein externer Fehler. Zu diesem externen Fehler liegen weitere Diagnoseinformationen vor, die mit der Anweisung „RDSYSST“ ausgelesen werden können. Wenn ein Kanalfehler vorhanden ist, ist die Variable „OB82_PNT_INFO“ des Diagnosealarm-OB (OB82) auf „1“ gesetzt. Daraufhin wird im OB82 der Merker M20.0 „channel_failure“ auf „1“ gesetzt. Wenn der M20.0 „channel_failure“ auf „1“ gesetzt ist, wird der FB30 „FB_chaF“ im OB1 aufgerufen und ausgeführt. Im FB30 „FB_chaF“ wird die Anweisung „RDSYSST“ aufgerufen, um die Diagnoseinformationen auszulesen. Wenn die Diagnoseinformationen ausgelesen wurden, wird der M20.0 „channel_failure“ wieder zurückgesetzt.

Abbildung 3-3

```
L      4                //module address of the digital input module
L      #OB82_MDL_ADDR
==I
SPBN  END                // if the module address of the OB82 is 4, do not jump

// here we do everything for module on address 4
// b#16#39 for entering the diagnostic, b#16#38 for leaving the diagnostic

L      b#16#39
L      #OB82_EV_CLASS
==I
SPBN  MOOK                //if no entering diagnostic > module is ok

//here we do everything for module on address 4 an entering diagnostic

U      #OB82_EXT_FAULT
S      "external_failure" //external fault marker
SPBN  NEXF

U      #OB82_INT_FAULT
S      "internal_failure"

U      #OB82_PNT_INFO
SPBN  END                //jump not if there is more channel informations available

S      "channel_failure"
SPA   END

NEXF: NOP 0
//here you may react to internal faults
SET
=      "internal_failure"
SPA   END

MOOK: NOP 0
//reset all markers if no entering diagnostic > the module is ok
CLR
=      "channel_failure"
=      "internal_failure"
=      "external_failure"
SPA   END

END:  NOP 0
BEA
```

3.3 OB1

Wenn Merker M20.0 „channel_failure“ auf „1“ gesetzt ist, wird der Funktionsbaustein FB30 „FB_chaF“ aufgerufen und ausgeführt.

3.4 FB30 „FB_chaF“

3.4.1 Diagnosedaten mit der Systemfunktion SFC51 „RDSYSST“ auslesen

In diesem Beispiel wird die Anweisung „RDSYSST“ im FB30 „FB_chaF“ mit der SZL-ID w#16#B3 aufgerufen und es wird auf die digitale Eingabebaugruppe mit der Modul-Adresse 4 zugegriffen, die den Diagnosealarm gestellt hat. Damit wird der Lesevorgang sofort ausgeführt.

Abbildung 3-4

```

CALL RDSYSST
  REQ      :=TRUE
  SZL_ID   :=w#16#00B3
  INDEX    :=w#16#4
  RET_VAL  :="Tag_3"
  BUSY     :="Tag_4"
  SZL_HEADER :="DB_SZL_Header".SZL_HEADER
  DR       :=P#DB1.DBX0.0 BYTE 100
    
```

Eingangsparameter der Anweisung „RDSYSST“

Die Anweisung „RDSYSST“ hat folgende Eingangsparameter“.

Tabelle 3-1

Eingangsparameter	Datentyp	Beschreibung
REQ	BOOL	REQ=1: Anstoß der Bearbeitung
SZL_ID	WORD	SZL-ID der Teilliste oder des Teillistenauszugs In diesem Beispiel wird die SZL-ID w#16#B3 verwendet. Mit dieser SZL-ID werden alle Diagnosedaten einer Baugruppe (Diagnosedatensatz DS1) gelesen. Eine Übersicht der SZL_IDs für die S7-300 CPUs und ET 200-CPU's finden Sie in der Operationsliste unter der Beitrags-ID 31977679 . Eine Übersicht der SZL-ID's für die S7-400 CPUs finden Sie in der STEP 7 Online-Hilfe sowie im Handbuch „STEP 7 Professional V13“ unter der Beitrags-ID 89515142 .
INDEX	WORD	Typ oder Nummer eines Objekts in einer Teilliste Wenn Sie die SZL-ID w#16#B3 auslesen geben Sie am Parameter INDEX die Modul-Adresse, in diesem Beispiel die Anfangsadresse der Eingänge, an.

Ausgangsparameter der Anweisung „RDSYSST“

Die Anweisung „RDSYSST“ hat folgende Ausgangsparameter“.

Tabelle 3-2

Ausgangsparameter	Datentyp	Beschreibung
RET_VAL	INT	Tritt während der Bearbeitung der Anweisung ein Fehler auf, dann enthält der Parameter RET_VAL einen Fehlercode.
BUSY	BOOL	TRUE: Lesevorgang noch nicht abgeschlossen
SZL_HEADER	STRUCT	Der Aufbau der Datenstruktur „SZL_HEADER“ ist im Kapitel beschrieben.
DR	ANY	Zielbereich für die gelesene SZL-Teilliste bzw. den gelesenen SZL-Teillistenauszug: <ul style="list-style-type: none"> Falls Sie nur die Kopfinformation einer SZL-Teilliste ausgelesen haben, dürfen Sie DR nicht auswerten, sondern nur SZL_HEADER. Andernfalls gibt das Produkt aus LENTHDR und N_DR an, wie viele Bytes in DR eingetragen wurden. Der Aufbau des Zielbereichs ist im Kapitel beschrieben.

3.4.2 Aufbau der Datenstruktur „SZL_HEADER“

Die folgende Abbildung zeigt den Aufbau der Datenstruktur „SZL_HEADER“.

Abbildung 3-5

DB_SZL_Header				
	Name	Datentyp	Offset	Startwert
1	Static			
2	SZL_HEADER	Struct	0.0	
3	LENTHDR	Word	0.0	16#0
4	N_DR	Word	2.0	16#0

LENTHDR ist die Länge eines Datensatzes der SZL-Teilliste oder des SZL-Teillistenauszugs.

- Falls Sie nur die Kopfinformation einer SZL-Teilliste ausgelesen haben, enthält N_DR die Anzahl der vorhandenen zugehörigen Datensätze.
- Andernfalls enthält N_DR die Anzahl der in den Zielbereich übertragenen Datensätze.

3.4.3 Aufbau der Diagnosedaten

[Tabelle 3-3](#) zeigt den Aufbau und Inhalt der Diagnosedaten einer Baugruppe.

Die Diagnosedaten einer Baugruppe stehen in den Datensätzen 0 und 1 des Systemdatenbereichs.

- Der Datensatz 0 enthält 4 Byte Diagnosedaten, die den aktuellen Zustand einer Signalbaugruppe beschreiben.
- Der Datensatz 1 enthält:
 - die 4 Byte Diagnosedaten, die auch im Datensatz 0 stehen, und
 - die baugruppenspezifischen Diagnosedaten.

Wenn ein Fehler auftritt, dann wird das entsprechende Bit auf „1“ gesetzt.

Tabelle 3-3

Byte	Bit	Bedeutung	Datensatz
0	0	Baugruppenstörung	0 und 1
	1	Fehler intern	
	2	Fehler extern	
	3	Kanalfehler vorhanden	
	4	Externe Hilfsspannung fehlt	
	5	Frontstecker fehlt	
	6	Parametrierung fehlt	
	7	Falsche Parameter in der Baugruppe	
1	0-3	Baugruppenklasse: 0101: Analogbaugruppe 0000: CPU 1000: Funktionsbaugruppe 1100: CP 1111: Digitalbaugruppe 0011: DP-Normslave 1011: I-Slave 0100: IM	0 und 1
	4	Kanalinformation vorhanden	
	5	Anwenderinformation vorhanden	
	6	Diagnosealarm von Stellvertreter	
	7	Wartungsbedarf (nur bei PROFINET IO)	
2	0	Speichermodul falsch oder fehlt	0 und 1
	1	Kommunikationsstörung	
	2	Betriebszustand	
	3	Zykluszeitüberwachung angesprochen	
	4	Baugruppeninterne Versorgungsspannung ausgefallen	
	5	Batterie leer	
	6	Gesamte Pufferung ausgefallen	
	7	Wartungsanforderung (nur bei PROFINET IO)	
3	0	Erweiterungsgeräteausfall	0 und 1
	1	Prozessorausfall	
	2	EPROM-Fehler	
	3	RAM-Fehler	
	4	ADU/DAU-Fehler	
	5	Sicherungsausfall	
	6	Prozessalarm verloren	
	7	Reserviert	
4	0-6	Kanaltyp: B#16#70: Digitaleingabe B#16#72: Digitalausgabe B#16#71: Analogeingabe B#16#73: Analogausgabe B#16#74: FM-POS B#16#75: FM-REG	1

Byte	Bit	Bedeutung	Datensatz
		B#16#76: FM-ZAEHL B#16#77: FM-TECHNO B#16#78: FM-NCU B#16#79 bis B#16#7D: reserviert B#16#7E: US300 B#16#7F: reserviert	
	7	Weiterer Kanaltyp vorhanden: 0: nein 1: ja	
5	0-7	Anzahl der Diagnosebits, die eine Baugruppe pro Kanal ausgibt	1
6	0-7	Anzahl der gleichartigen Kanäle einer Baugruppe	1
7	0	Kanalfehler Kanal 0 / Kanalgruppe 0 (z. B. Digitaleingabekanal 0 und 1)	1
	1	Kanalfehler Kanal 1 / Kanalgruppe 1 (z. B. Digitaleingabekanal 2 und 3)	
	2	Kanalfehler Kanal 2 / Kanalgruppe 2 (z. B. Digitaleingabekanal 4 und 5)	
	3	Kanalfehler Kanal 3 / Kanalgruppe 3 (z. B. Digitaleingabekanal 6 und 7)	
	4	Kanalfehler Kanal 4 / Kanalgruppe 4 (z. B. Digitaleingabekanal 8 und 9)	
	5	Kanalfehler Kanal 5 / Kanalgruppe 5 (z. B. Digitaleingabekanal 10 und 11)	
	6	Kanalfehler Kanal 6 / Kanalgruppe 6 (z. B. Digitaleingabekanal 12 und 13)	
	7	Kanalfehler Kanal 7 / Kanalgruppe 7 (z. B. Digitaleingabekanal 14 und 15)	
...		kanalspezifische Diagnosedaten (siehe Tabelle 3-4)	1

Aufbau der kanalspezifischen Diagnosedaten

[Tabelle 3-4](#) zeigt den Aufbau der kanalspezifischen Diagnosedaten. Die kanalspezifischen Diagnosedaten enthalten u. a. die Information zum Drahtbruch auf einem Kanal oder einer Kanalgruppe. Als Beispiel wird der Aufbau des Diagnosebyte für einen Digitaleingabekanal gezeigt.

Tabelle 3-4

Bit	Bedeutung	Bemerkung
0	Projektierungs-/Parametrierfehler	meldbar mit der Anweisung „WR_USMSG“ (SFC52) und EVENTN=w#16#8x70
1	Massefehler	meldbar mit der Anweisung „WR_USMSG“ (SFC52) und EVENTN=w#16#8x71
2	P-Kurzschluss (Geber)	meldbar mit der Anweisung „WR_USMSG“ (SFC52) und EVENTN=w#16#8x72
3	M-Kurzschluss	meldbar mit der Anweisung


Bit	Bedeutung	Bemerkung
		„WR_USMSG“ (SFC52) und EVENTN=w#16#8x73
4	Drahtbruch	meldbar mit der Anweisung „WR_USMSG“ (SFC52) und EVENTN=w#16#8x74
5	Geberversorgung fehlt	meldbar mit der Anweisung „WR_USMSG“ (SFC52) und EVENTN=w#16#8x75
6	0	
7	0	

Um einen Drahtbruch auf einem Digitaleingabekanal im Anwenderprogramm zu erkennen, werten Sie das Bit 4 des Diagnosebyte aus. Wenn das Bit 4 auf „1“ gesetzt ist, liegt ein Drahtbruch vor.

3.4.4 Drahtbruch auf Kanal 0 und/oder Kanal 1 einer digitalen Eingabebaugruppe erkennen

Die Diagnosedaten werden im Datenbaustein DB1 gespeichert. In diesem Beispiel sind die ausgelesenen Diagnosedaten (Datensatz 1) insgesamt 16 Byte lang. Diese Information finden Sie in der Datenstruktur „SZL_HEADER“ (siehe [Abbildung 3-6](#)).

Abbildung 3-6



Die kanalspezifischen Diagnosedaten umfassen 8 Byte und sind im Datenbaustein DB1 ab Byte 8 gespeichert.

Tabelle 3-5

Datenbaustein DB1	Beschreibung
Byte 0-7	Diagnosebyte 0 bis 7 des Datensatz 1
Byte 8	Diagnosebyte für Digitaleingabekanal 0 und 1
Byte 9	Diagnosebyte für Digitaleingabekanal 2 und 3
Byte 10	Diagnosebyte für Digitaleingabekanal 4 und 5
Byte 11	Diagnosebyte für Digitaleingabekanal 6 und 7
Byte 12	Diagnosebyte für Digitaleingabekanal 8 und 9
Byte 13	Diagnosebyte für Digitaleingabekanal 10 und 11
Byte 14	Diagnosebyte für Digitaleingabekanal 12 und 13
Byte 15	Diagnosebyte für Digitaleingabekanal 14 und 15

Um einen Drahtbruch auf Kanal 0 und/oder Kanal 1 einer digitalen Eingabebaugruppe zu erkennen, werten Sie das Bit 0 im Byte 7 und das Bit 4 im Byte 8 der ausgelesenen Diagnosedaten (Datensatz 1) aus. Wenn die beiden Bits auf „1“ gesetzt sind, dann liegt ein Drahtbruch auf Kanal 0 und/oder Kanal 1 vor. In diesem Fall wird das anwenderdefinierte Diagnoseereignis „Drahtbruch“ in den Diagnosepuffer der CPU geschrieben (siehe). Das Datenwort #INFO1 wird auf den Wert 0 gesetzt (für Kanal 0) und das Datenwort #INFO2 wird auf den Wert 1 gesetzt (für Kanal 1).

Abbildung 3-7

```

U      %DB1.DBX7.0
U      %DB1.DBX8.4

S      "wire_break_0_1"

U      "wire_break_0_1"
SPBN  FBCE

L      0
T      #INFO1

L      1
T      #INFO2

CALL  WR_USMSG
SEND  :=TRUE
EVENTN :=w#16#8174
INFO1  :=#INFO1
INFO2  :=#INFO2
RET_VAL :=#varRET_VAL
    
```

3.4.5 SFC52 „WR_USMSG“

Mit der Systemfunktion SFC52 „WR_USMSG“ (write user element in diagnostic buffer) schreiben Sie ein anwenderdefiniertes Diagnoseereignis in den Diagnosepuffer. Zusätzlich können Sie die zugehörige Diagnosemeldung an alle dafür angemeldeten Teilnehmer senden (durch Belegung des Eingangsparameters SEND = TRUE).“

Abbildung 3-8

```

CALL  WR_USMSG
SEND  :="wire_break_0_1"
EVENTN :=w#16#8174
INFO1  :=#INFO1
INFO2  :=#INFO2
RET_VAL :=#varRET_VAL
    
```

Die Anweisung „WR_USMSG“ hat folgende Eingangsparameter:

Tabelle 3-6

Eingangsparameter	Datentyp	Beschreibung
SEND	BOOL	Senden der anwenderdefinierten Diagnosemeldung an alle angemeldeten Teilnehmer freigeben.
EVENTN	WORD	Ereignis-ID

Eingangsparameter	Datentyp	Beschreibung
		Mit der Ereignis-ID w#16#8174 wird der Drahtbruch als kommendes Ereignis im Diagnosepuffer der CPU eingetragen.
INFO1	ANY	Zusatzinformation 1 Wort lang
INFO2	ANY	Zusatzinformation 2 Wort lang

Die folgende Abbildung zeigt das anwenderdefinierte Ereignis, das bei einem Drahtbruch auf Kanal 0 und/oder Kanal 1 in den Diagnosepuffer der CPU geschrieben wird.

Abbildung 3-9

Diagnosepuffer

Ereignisse

CPU-Zeitstempel berücksichtigt lokale PG/PC-Zeit

Nr.	Datum und Uhrzeit	Ereignis
1	05.09.2014 14:42:05.59	Digitaleingabe: Drahtbruch: gekommen
2	05.09.2014 14:42:05.58	Baugruppe gestört oder Wartung erforderlich
3	05.09.2014 14:41:31.20	Betriebszustandsübergang von ANLAUF nach RUN
4	05.09.2014 14:41:31.20	Manuelle Neustart (Warmstart)-Anforderung
5	05.09.2014 14:41:31.15	Betriebszustandsübergang von STOP nach ANLAUF
6	05.09.2014 14:41:31.15	Neue Anlaufinformation im Betriebszustand STOP
7	05.09.2014 14:41:24.96	Neue Anlaufinformation im Betriebszustand STOP
8	05.09.2014 14:41:24.96	Neue Anlaufinformation im Betriebszustand STOP

Anzeige einfrieren

Details zum Ereignis

Details zum Ereignis: 1 von 10 Ereignis-ID: 16# 8174

Beschreibung: Digitaleingabe: Drahtbruch: gekommen

Zeitstempel: 05.09.2014 14:42:05.592

Kommend/Gehend: Kommendes Ereignis

Hilfe zum Ereignis Im Editor öffnen Speichern unter...

3.5 FB20 „FB_extF“

Der FB20 „FB_extF“ wird im OB1 aufgerufen und ausgeführt, wenn ein externer Fehler von der digitalen Eingabebaugruppe gemeldet wird. Hier können Sie entsprechend auf externe Fehler reagieren.

3.6 FB10 „FB_intF“

Der FB10 „FB_intF“ wird im OB1 aufgerufen und ausgeführt, wenn ein interner Fehler von der digitalen Eingabebaugruppe gemeldet wird. Hier können Sie entsprechend auf interne Fehler reagieren.